9. přednáška (extrémy, slovní úlohy vedoucí na extrémy)

Lokální extrémy

V Maplu lze lokální extrémy dané funkce najít pomocí příkazu extrema.

Příklad:

Najděte lokální extrémy funkce $f(x) = x^3 - 12\,x + 1$.

```
> restart;
> f:=x^3-12*x+1;
```

$$f := x^3 - 12\,x + 1$$

```
> extrema(f,{},x);
```

$$\{-15, 17\}$$

```
> plot(f,x=-4..4,thickness=5);
```



Vidíme, že příkaz extrema vypsal pouze funkční hodnoty v lokálních extrémech.

```
> extrema(f,{},x,'bod');
```

$$\{-15, 17\}$$

```
> bod;
```

$$\{\{x = -2\}, \{x = 2\}\}$$

Dále si můžeme všimnout, že Maple nevypsal, zda se jedná o maximum nebo minimum.

```
>
```

Při bližším zkoumání zjistíme, že příkaz extrema ve skutečnosti nehledá lokální extrémy, ale pouze stacionární body :-(což ostatně ukazuje následující příklad.

```
> extrema((x-1)^3,{},x,'bod');
```

$$\{0\}$$

```
> bod;
```

$$\{\{x=1\}\}$$

Funkce $f(x)=(x-1)^3$ přitom žádný lokální extrém nemá!

```
> plot((x-1)^3,x=0..2,thickness=5);
```




```
>
```

Možná by bylo lepší použít příkaz ExtremePoints z balíku Student[Calculus1].

```
> with(Student[Calculus1]);
```

[*AntiderivativePlot*, *AntiderivativeTutor*, *ApproximateInt*, *ApproximateIntTutor*, *ArcLength*, *ArcLengthTutor*, *Asymptotes*, *Clear*, *CriticalPoints*, *CurveAnalysisTutor*, *DerivativePlot*, *DerivativeTutor*, *DiffTutor*, *ExtremePoints*, *FunctionAverage*, *FunctionAverageTutor*, *FunctionChart*, *FunctionPlot*, *GetMessage*, *GetNumProblems*, *GetProblem*, *Hint*, *InflectionPoints*, *IntTutor*, *Integrand*, *InversePlot*, *InverseTutor*, *LimitTutor*, *MeanValueTheorem*, *MeanValueTheoremTutor*, *NewtonQuotient*, *NewtonsMethod*, *NewtonsMethodTutor*, *PointInterpolation*, *RiemannSum*, *RollesTheorem*, *Roots*, *Rule*, *Show*, *ShowIncomplete*, *ShowSolution*, *ShowSteps*, *Summand*, *SurfaceOfRevolution*, *SurfaceOfRevolutionTutor*, *Tangent*, *TangentSecantTutor*, *TangentTutor*, *TaylorApproximation*, *TaylorApproximationTutor*, *Understand*, *Undo*, *VolumeOfRevolution*, *VolumeOfRevolutionTutor*, *WhatProblem*]

```
> ExtremePoints((x-1)^3);
```

$$[\ ]$$

Vidíme, že nyní Maple správně poznal, že funkce $f(x)=(x-1)^3$ žádný lokální extrém nemá. Nevýhodou je, že příkaz ExtremePoints funguje pouze pro funkce jedné proměnné.

```
> CriticalPoints((x-1)^3);
```

$$[1]$$

```
>
```

Příklad:

Najděte lokální extrémy funkce $f(x, y) = x^2 - y^3 + 7\,y^2 - 4\,x\,y - 4\,x + 17\,y + 11$.

```
> f:=x^2-y^3+7*y^2-4*x*y-4*x+17*y+11;
```

$$f := x^2 - y^3 + 7\,y^2 - 4\,x\,y - 4\,x + 17\,y + 11$$

```
> extrema(f,{},{x,y},bod);
Error, (in extrema) illegal use of a formal parameter
>
```

Maple vypsal chybovou hlášku, protože v proměnné "bod" už je přiřazená hodnota (ještě z předchozího příkladu).

```
> bod;
```

$$\{\{x = 1\}\}$$

```
>
```

Abychom zamezili výskytu chybové hlášky, použijeme apostrofy.

```
> extrema(f,{},{x,y},'bod');
```

$$\{2, 34\}$$

```
> bod;
```

$$\{\{x = 0, y = -1\}, \{x = 8, y = 3\}\}$$

Již víme, že to, co vidíme, nejsou extrémy, ale pouze stacionární body, takže zbytek musíme dodělat sami.
Sestavíme matici druhých derivací.

```
> H:=unapply(<<diff(f,x,x),diff(f,y,x)>|<diff(f,x,y),diff(f,y,y)>>,x,y);
```

$$H := (x, y) \rightarrow \mathrm{rtable}(1 \,..\, 2, 1 \,..\, 2, \{(1, 1) = 2, (1, 2) = -4, (2, 1) = -4, (2, 2) = -6\,y + 14\}, \\ \mathit{datatype} = \mathit{anything}, \mathit{subtype} = \mathit{Matrix}, \mathit{storage} = \mathit{rectangular}, \mathit{order} = \mathit{Fortran\_order})$$

```
> H(x,y);
```

$$\begin{bmatrix} 2 & -4 \\ -4 & -6\,y + 14 \end{bmatrix}$$

```
> H(8,3);
```

$$\begin{bmatrix} 2 & -4 \\ -4 & -4 \end{bmatrix}$$

```
> LinearAlgebra[IsDefinite](H(8,3),query='indefinite');
```

$$true$$

Vidíme, že matice $H(8,3)$ je indefinitní, a proto v bodě $(8, 3)$ lokální extrém nenastává.

```
> H(0,-1);
```

$$\begin{bmatrix} 2 & -4 \\ -4 & 20 \end{bmatrix}$$

```
> LinearAlgebra[IsDefinite](H(0,-1));
```

$$true$$

```
> LinearAlgebra[IsDefinite](H(0,-1),query='positive_definite');
```

$$true$$

Matice $H(0,-1)$ je pozitivně definitní, a proto v bodě $(0, -1)$ nastane ostré lokální minimum.

```
>
```

Graf funkce f:

```
> plot3d(f,x=-7..9,y=-4..5,axes=framed,orientation=[120,60]);
```




```
>
```

Graf funkce f v okolí stacionárního bodu $(8, 3)$:

```
> plot3d(f,x=6..10,y=1..4,axes=framed,orientation=[-150,75]);
```



>

Graf funkce f v okolí stacionárního bodu $(0, -1)$:

```
> plot3d(f,x=-5..5,y=-3..2,axes=framed,orientation=[120,60]);
```



>

Vrstevnice funkce f:

```
> with(plots):
> p1:=plots[contourplot](f,x=-7..12,y=-4..8,contours=[seq(3+i*2
  0,i=-30..30),34,2,9],grid=[60,60],thickness=3):
> p2:=pointplot([[8,3],[0,-1]],symbol=soliddiamond,symbolsize=2
  0,color=black):
```

```
> display([p1,p2]);
```




```
>
```

Globální extrémy

Pro výpočet globálních extrémů v Maplu slouží příkazy maximize a minimize.

Připomeňme jednu důležitou větu z matematické analýzy.

Weierstrassova věta:
Nechť funkce f (obecně n proměnných) je spojitá na neprázdné kompaktní množině M. Pak funkce f nabývá na M svého (globálního) maxima i minima.

Příklad:

Najděte globální extrémy funkce $\mathrm{f}(x) = x^3 - 12\,x + 1$ na intervalu <–3, 5>.

```
> restart;
> f:=x^3-12*x+1;
```

$$f := x^3 - 12\,x + 1$$

```
> maximize(f,x=-3..5);
```

$$66$$

```
> minimize(f,x=-3..5);
```

$$-15$$

Dostali jsme pouze hodnotu maxima a minima - ne však body, ve kterých maximum či minimum nastává.

```
> maximize(f,x=-3..5,location);
```

$$66, \{[\{x=5\}, 66]\}$$

```
> minimize(f,x=-3..5,location);
```

$$-15, \{[\{x=2\}, -15]\}$$

```
> plot(f,x=-3..5,thickness=5);
```




```
>
```

Příklad:

Najděte globální extrémy funkce $f(x, y) = x^2 + x\,y + y^2$ na čtverci <0, 1> x <0, 1>.

```
> f:=x^2+x*y+y^2;
```

$$f := x^2 + x\,y + y^2$$

```
> minimize(f,x=0..1,y=0..1,location);
```

$$0, \{[\{x=0, y=0\}, 0]\}$$

```
> maximize(f,x=0..1,y=0..1,location);
```

$$3, \{[\{x=1, y=1\}, 3]\}$$

```
>
```

Graf funkce *f*:

```
> plot3d(f,x=0..1,y=0..1,axes=framed,orientation=[-55,75]);
```



>

Vrstevnice funkce *f*:

```
> with(plots):
> contourplot(f,x=0..1,y=0..1,contours=40,coloring=[blue,red],thickness=3);
```



>

Problém nastává, pokud bychom například chtěli počítat maximum nebo minimum funkce dvou proměnných na obecnějších množinách, než jsou dvojrozměrné intervaly.

>

Příklad:

Najděte globální extrémy funkce $\mathrm{f}(x, y) = x^2 + x\,y + y^2$ na kruhu $x^2 + y^2 \le 1$.

```
> f:=(x,y)->x^2+x*y+y^2;
```

$$f := (x, y) \to x^2 + x\,y + y^2$$

```
> minimize(f(x,y),y=-sqrt(1-x^2)..sqrt(1-x^2),x=-1..1,location)
  ;
```

$$\mathrm{minimize}(x^2 + x\,y + y^2, y = -\sqrt{1-x^2} \,..\, \sqrt{1-x^2}, x = -1 \,..\, 1, \mathit{location}), \{\ \}$$

Zde si musíme poradit opět sami.

Nejprve vypočteme stacionární body.

```
> extrema(f(x,y),{},{x,y},'bod');
```

$$\{0\}$$

```
> bod;
```

$$\{\{x = 0, y = 0\}\}$$

Dostali jsme stacionární bod (0, 0), který leží uvnitř zadané množiny.
Nyní je potřeba vyšetřit hranici kruhu. Tuto hranici si rozdělíme na dva kusy (horní a dolní půlkružnice).

Horní půlkružnice:

```
> h:=x->sqrt(1-x^2);
```

$$h := x \to \sqrt{1-x^2}$$

```
> minimize(f(x,h(x)),x=-1..1,location);
```

$$\frac{1}{2}, \left\{\left[\left\{x = -\frac{\sqrt{2}}{2}\right\}, \frac{1}{2}\right]\right\}$$

y-ová souřadnice:

```
> y=h(-sqrt(2)/2);
```

$$y = \frac{\sqrt{2}}{2}$$

```
> maximize(f(x,h(x)),x=-1..1,location);
```

$$\frac{3}{2}, \left\{\left[\left\{x = \frac{\sqrt{2}}{2}\right\}, \frac{3}{2}\right]\right\}$$

y-ová souřadnice:

```
> y=h(sqrt(2)/2);
```

$$y = \frac{\sqrt{2}}{2}$$

>

Dolní půlkružnice:

```
> h:=x->-sqrt(1-x^2);
```

$$h := x \to -\sqrt{1 - x^2}$$

```
> minimize(f(x,h(x)),x=-1..1,location);
```

$$\frac{1}{2}, \left\{\left[\left\{x = \frac{\sqrt{2}}{2}\right\}, \frac{1}{2}\right]\right\}$$

y-ová souřadnice:

```
> y=h(sqrt(2)/2);
```

$$y = -\frac{\sqrt{2}}{2}$$

```
> maximize(f(x,-sqrt(1-x^2)),x=-1..1,location);
```

$$\frac{3}{2}, \left\{\left[\left\{x = -\frac{\sqrt{2}}{2}\right\}, \frac{3}{2}\right]\right\}$$

y-ová souřadnice:

```
> y=h(-sqrt(2)/2);
```

$$y = -\frac{\sqrt{2}}{2}$$

Hodnota ve stacionárním bodě:

```
> f(0,0);
```

$$0$$

Odtud vidíme, že globální minimum nastává v bodě $(0, 0)$ a globální maximum nastává v bodech $\left(\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{2}\right)$ a $\left(-\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2}\right)$. Hodnota globálního minima je 0 a hodnota globálního maxima je $\frac{3}{2}$.

>

Graf funkce f:

```
> p1:=plot3d(f(x,y),y=-sqrt(1-x^2)..sqrt(1-x^2),x=-1..1,axes=fr
  amed):
> p2:=spacecurve([cos(t),sin(t),0],t=0..2*Pi,color=black,thickn
  ess=5):
> display([p1,p2],orientation=[30,60]);
```




```
>
```

Vrstevnice funkce *f*:

```
> p1:=contourplot(f(x,y),y=-sqrt(1-x^2)..sqrt(1-x^2),x=-1..1,co
  ntours=15,coloring=[blue,red],thickness=3):
> p2:=plot([cos(t),sin(t),t=0..2*Pi],color=black,thickness=5):
> display([p1,p2]);
```



Jinou možnost dávají příkazy Maximize a Minimize z balíku Optimization.

Výhoda je, že tyto příkazy se umí vypořádat s obecnými množinami, ale bohužel Maple výpočet provádí numericky.

```
> with(Optimization);
```

[ImportMPS, Interactive, LPSolve, LSSolve, Maximize, Minimize, NLPSolve, QPSolve]

```
> Maximize(f(x,y),{x^2+y^2<=1});
```

$$[1.50000000000338306, [x = 0.707106781187345, y = 0.707106781187345]]$$

```
> Minimize(f(x,y),{x^2+y^2<=1});
```

$$[0.551012976947947269\ 10^{-39}, [x = -0.135525271560688\ 10^{-19}, y = -0.135525271560688\ 10^{-19}]]$$

```
>
```

Zároveň vidíme, že Maple nenašel všechny body, ve kterých nastává maximum.

```
>
```

Extremální úlohy

Krabice

Ze čtvercového kartónu o straně 60 cm odřízneme v rozích 4 menší čtverce. Ze zbytku poskládáme krabici bez víka (viz obrázek). Jak velké čtverce musíme v rozích odstřihnout, abychom dostali krabici o maximálním objemu. Jak velký bude onen maximální objem?



Označme x délku stran čtverců, které v rozích odřízneme. Pak objem krabice je popsán funkcí $\mathrm{V}(x) = (60 - 2\,x)^2\,x$. Proměnná x přitom může probíhat interval <0, 30>.
Úlohu jsme tedy převedli na výpočet **globálního** maxima dané funkce na daném intervalu.

```
> restart;
> V:=x->(60-2*x)^2*x;
```

$$V := x \rightarrow (60 - 2\,x)^2\,x$$

```
> maximize(V(x),x=0..30,location);
```

$$16000, \{[\{x = 10\}, 16000]\}$$

Vidíme, že musíme v rozích odstřihnout čtverce o straně 10 cm. Maximální objem krabice pak bude 16000 krychlových centimetrů, tj. 16 litrů.

```
>
```

Poznámka:

Kdybychom úlohu chtěli zobecnit a uvažovat čtvercový kartón o straně a, funkce V by měla předpis $\mathrm{V}(x) = (a - 2\,x)^2\,x$ a proměnná x by mohla být v intervalu $<0, \frac{a}{2}>$.
Funkce V je na tomto (uzavřeném) intervalu spojitá, a proto (podle Weierstrassovy věty) globální extrémy existují.

```
> V:=x->(a-2*x)^2*x;
```

$$V := x \rightarrow (a - 2\,x)^2\,x$$

```
> maximize(V(x),x=0..a/2,location);
```

$$\mathrm{maximize}\left((a - 2\,x)^2\,x, x = 0\,..\,\frac{a}{2}, location\right), \{\ \}$$

Zde už má Maple zase problémy.

```
> extrema(V(x),{},x,'bod');
```

$$\{\max\left(0, \frac{2\,a^3}{27}\right), \min\left(0, \frac{2\,a^3}{27}\right)\}$$

```
> bod;
```

$$\{\{x=\frac{a}{2}\}, \{x=\frac{a}{6}\}\}$$

Máme jediný stacionární bod $x=\frac{a}{6}$ ležící uvnitř intervalu $<0, \frac{a}{2}>$. Dále jsou samozřejmě "podezřelé" krajní body, tzn. $x=0$ a $x=\frac{a}{2}$.
Nyní už jen musíme porovnat funkční hodnoty v podezřelých bodech.

```
> V(0);
```

$$0$$

```
> V(a/2);
```

$$0$$

```
> V(a/6);
```

$$\frac{2\,a^3}{27}$$

Vzhledem k tomu, že a byl kladný parametr, globální maximum nastává v bodě $\frac{a}{6}$.
Musíme tedy v rozích odstřihnout menší čtverce o straně $\frac{a}{6}$, kde a byla strana původního čtverce.

```
>
```

Válec

Ze všech válců o daném objemu V najděte ten, který má minimální povrch. Jaké budou rozměry hledaného válce?

Poloměr válce si označme r a výšku v.

```
> restart;
> V=Pi*r^2*v;
```

$$V=\pi\,r^2\,v$$

```
> v:=solve(%,v);
```

$$v := \frac{V}{\pi\, r^2}$$

```
> S:=unapply(2*Pi*r^2+2*Pi*r*v,r);
```

$$S := r \rightarrow 2\,\pi\, r^2 + \frac{2\,V}{r}$$

```
>
```

Dostali jsme funkci, kterou je třeba minimalizovat. Zde hledáme globální minimum funkce S na intervalu (0, *nekonečno*).

Stacionární bod(y) funkce S:

```
> extrema(S(r),{},r,'stac');
```

$$\{ 3\,\pi^{(1/3)}\, 2^{(1/3)}\, V^{(2/3)} \}$$

```
> stac;
```

$$\{ \{ r = \frac{4^{(1/3)}\, (V\,\pi^2)^{(1/3)}}{2\,\pi} \} \}$$

Derivace funkce S:

```
> D(S)(r);
```

$$4\,\pi\, r - \frac{2\,V}{r^2}$$

Vidíme, že nalevo od stacionárního bodu je derivace funkce S záporná, zatímco napravo je kladná, tzn., že ve stacionárním bodě nastává lokální, ale i globální minimum.

```
> assign(stac);
> r:=simplify(r);
```

$$r := \frac{2^{(2/3)}\, V^{(1/3)}}{2\,\pi^{(1/3)}}$$

```
> v:=simplify(v);
```

$$v := \frac{2^{(2/3)}\, V^{(1/3)}}{\pi^{(1/3)}}$$

```
> testeq(v=2*r);
```

$$\mathit{true}$$

Optimální válec má tedy výšku rovnou průměru.

```
>
```

Parabola

Na parabole $p(x) = x^2$ najděte bod, který je nejblíže bodu $(1, 0)$.

Obecný bod na parabole je (x, x^2), kde x může být libovolné reálné číslo.
Vzdálenost takového bodu od bodu $(1, 0)$ je:

```
> restart;
> vzdalenost:=sqrt((x-1)^2+(x^2-0)^2);
```

$$vzdalenost := \sqrt{x^2 - 2\,x + 1 + x^4}$$

```
> reseni:=minimize(vzdalenost,location);
```

$$reseni := \sqrt{1 + \frac{\left((54+6\sqrt{87})^{(2/3)} - 6\right)^2}{36\,(54+6\sqrt{87})^{(2/3)}} - \frac{(54+6\sqrt{87})^{(2/3)} - 6}{3\,(54+6\sqrt{87})^{(1/3)}} + \frac{\left((54+6\sqrt{87})^{(2/\ldots}\right.}{1296\,(54+6\sqrt{8\ldots}}}$$

$$, \left\{\left[\left\{x = \frac{(54+6\sqrt{87})^{(2/3)} - 6}{6\,(54+6\sqrt{87})^{(1/3)}}\right\}, \sqrt{1 + \frac{\left((54+6\sqrt{87})^{(2/3)} - 6\right)^2}{36\,(54+6\sqrt{87})^{(2/3)}} - \frac{(54+6\sqrt{87}}{3\,(54+6\sqrt{\ldots}}}\right.\right.$$

$$\left.\left.\right]\right\}$$

```
> reseni[2,1,1];
```

$$\left\{x = \frac{(54+6\sqrt{87})^{(2/3)} - 6}{6\,(54+6\sqrt{87})^{(1/3)}}\right\}$$

```
> assign(reseni[2,1,1]);
> x;
```

$$\frac{(54+6\sqrt{87})^{(2/3)} - 6}{6\,(54+6\sqrt{87})^{(1/3)}}$$

```
> bod:=[x,x^2];
```

$$bod := \left[\frac{(54+6\sqrt{87})^{(2/3)} - 6}{6\,(54+6\sqrt{87})^{(1/3)}}, \frac{\left((54+6\sqrt{87})^{(2/3)} - 6\right)^2}{36\,(54+6\sqrt{87})^{(2/3)}}\right]$$

```
> evalf(bod);
```

$$[0.5897545123, 0.3478103848]$$

Grafické znázornění:

```
> with(plots):with(plottools):
```

```
> p1:=plot(x->x^2,-1..1,color=red,thickness=5):
> p2:=pointplot([bod,[1,0]],symbolsize=20,symbol=solidcircle
  ,color=black):
> p3:=line(bod,[1,0],color=black,thickness=5):
> display([p1,p2,p3],scaling=constrained);
```




```
>
```

Odlitek z betonu

Z 1 krychlového metru betonu máme odlít co nejvyšší těleso buď ve tvaru krychle, nebo koule, nebo koule postavené na krychli.
Jak vysoké bude toto těleso?

Označme x hranu krychle, kterou z betonu odlijeme. Je jasné, že x patří do intervalu < 0, 1>.

```
> restart;
> V_krychle:=x^3;
```

$$V\_krychle := x^3$$

```
> V_koule:=1-x^3;
```

$$V\_koule := 1 - x^3$$

```
> solve(4/3*Pi*r^3=V_koule,r);
```

$$\frac{((6-6x^3)\pi^2)^{(1/3)}}{2\pi}, -\frac{((6-6x^3)\pi^2)^{(1/3)}}{4\pi} - \frac{\frac{1}{4} I \sqrt{3}\,((6-6x^3)\pi^2)^{(1/3)}}{\pi},$$
$$-\frac{((6-6x^3)\pi^2)^{(1/3)}}{4\pi} + \frac{\frac{1}{4} I \sqrt{3}\,((6-6x^3)\pi^2)^{(1/3)}}{\pi}$$

```
> r:=%[1];
```

$$r := \frac{((6-6x^3)\pi^2)^{(1/3)}}{2\pi}$$

```
> vyska:=x+2*r;
```

$$vyska := x + \frac{((6-6x^3)\pi^2)^{(1/3)}}{\pi}$$

```
> maximize(vyska,x=0..1);
```

$$\text{maximize}\left(x + \frac{((6-6x^3)\pi^2)^{(1/3)}}{\pi}, x = 0\,..\,1\right)$$

Maple má s maximalizací problémy. Mohli bychom použít příkaz Maximize z balíku Optimization a spočítat maximum alespoň numericky.

```
> Optimization[Maximize](vyska,{x>=0,x<=1},initialpoint={x=1
/2});
```

$$[1.78357616610593218, [x = 0.748779886288184]]$$

```
> assign(%[2]);
> x;
```

$$0.748779886288184$$

```
> evalf(r);
```

$$0.517398139715417$$

```
>
```

Vidíme, že je potřeba odlít krychli o hraně 0.748779886288184 a kouli s poloměrem 0.517398139715417. Vznikne těleso s výškou 1.78357616610593218.

Animace:

```
> with(plots):
> with(plottools):
> kresli:=proc(x) local p1,p2,r;
p1:=hexahedron([0,0,x/2],x/2,color=grey);
r:=evalf(surd((1-x^3)*3/(4*Pi),3));
p2:=sphere([0,0,r+x],r,color=grey);
display([p1,p2],axes=framed,scaling=constrained,orientatio
n=[-135,75,0]);
end;
```

kresli := **proc**(x)

local $p1, p2, r$;

$p1 := plottools{:}\text{-}hexahedron([0, 0, 1/2*x], 1/2*x, color = grey)$;

$r := \text{evalf}(\text{surd}(3/4*(1-x\hat{}3)/\pi, 3))$;

$p2 := plottools{:}\text{-}sphere([0, 0, r+x], r, color = grey)$;

$plots\text{:-}display([p1, p2], axes = framed, scaling = constrained,$
$orientation = [-135, 75, 0])$

end proc

```
> animate(kresli,[t],t=0..1,scaling=constrained,frames=50);
```

```
> x;
```

$$0.748779886288184$$

```
> kresli(x);
```



Dokázali bychom Maple popostrčit, aby maximalizaci spočítal (aniž bychom museli přistupovat k numerickému řešení)?

```
>
```

Dvě chodby a žebřík

Dvě chodby široké a a b metrů se křižují v pravém úhlu. Zjistěte maximální délku žebříku, který lze ve vodorovné poloze přenést z jedné chodby do druhé.



Není těžké si uvědomit, že maximální délka žebříku je rovna minimu délek všech možných červených úseček (viz obrázek výše). Funkce, která délku červené úsečky popisuje, má tvar $d(t) = \frac{a}{\sin(t)} + \frac{b}{\cos(t)}$, kde t je úhel, který svírá úsečka s osou x. Úhel t přitom probíhá otevřený interval $(0, \frac{\pi}{2})$.

```
> restart;
> assume(a>0,b>0);
> d:=t->a/sin(t)+b/cos(t);
```

$$d := t \rightarrow \frac{a}{\sin(t)} + \frac{b}{\cos(t)}$$

```
> derivace:=D(d)(t);
```

$$derivace := -\frac{a\sim \cos(t)}{\sin(t)^2} + \frac{b\sim \sin(t)}{\cos(t)^2}$$

```
> solve(derivace,t);
```

$$\arctan\left(\frac{(a\sim b\sim^2)^{(1/3)}}{b\sim}\right), \arctan\left(-\frac{(a\sim b\sim^2)^{(1/3)}}{2\, b\sim} + \frac{\frac{1}{2} I \sqrt{3}\, (a\sim b\sim^2)^{(1/3)}}{b\sim}\right),$$
$$-\arctan\left(\frac{(a\sim b\sim^2)^{(1/3)}}{2\, b\sim} + \frac{\frac{1}{2} I \sqrt{3}\, (a\sim b\sim^2)^{(1/3)}}{b\sim}\right)$$

```
> stac_bod:=%[1];
```

$$stac\_bod := \arctan\left(\frac{(a\sim b\sim^2)^{(1/3)}}{b\sim}\right)$$

Funkce d má tedy jediný stacionární bod v intervalu $(0, \frac{\pi}{2})$. Zároveň je jasné, že v tomto otevřeném intervalu je funkce d diferencovatelná. Dále platí:

```
> Limit(d(x),x=0,right)=limit(d(x),x=0,right);
```

$$\lim_{x \to 0+} \frac{a\sim}{\sin(x)} + \frac{b\sim}{\cos(x)} = \infty$$

```
> Limit(d(x),x=Pi/2,left)=limit(d(x),x=Pi/2,left);
```

$$\lim_{x \to \left(\frac{\pi}{2}\right)-} \frac{a\sim}{\sin(x)} + \frac{b\sim}{\cos(x)} = \infty$$

```
>
```

Na základě toho není těžké si uvědomit, že funkce d má ve vypočteném stacionárním bodě **globální** minimum. Promyslete si to!

Hledaná délka:

```
> d(stac_bod);
```

$$\frac{a\sim b\sim \sqrt{1 + \frac{(a\sim b\sim^2)^{(2/3)}}{b\sim^2}}}{(a\sim b\sim^2)^{(1/3)}} + b\sim \sqrt{1 + \frac{(a\sim b\sim^2)^{(2/3)}}{b\sim^2}}$$

```
> delka:=simplify(%);
```

$$delka := \frac{\sqrt{b\sim^{(2/3)} + a\sim^{(2/3)}}\,(a\sim + b\sim^{(2/3)}\,a\sim^{(1/3)})}{a\sim^{(1/3)}}$$

```
> delka:=subs({a='a',b='b'},simplify(%));
```

$$delka := \frac{\sqrt{b^{(2/3)} + a^{(2/3)}}\,(a + b^{(2/3)}\,a^{(1/3)})}{a^{(1/3)}}$$

```
>
```

Hledaná maximální délka žebříku je $\left(a^{\left(\frac{2}{3}\right)} + b^{\left(\frac{2}{3}\right)}\right)^{\left(\frac{3}{2}\right)}$.

Skutečně:

```
> delka_upravena:=(a^(2/3)+b^(2/3))^(3/2);
```

$$delka\_upravena := (b\sim^{(2/3)} + a\sim^{(2/3)})^{(3/2)}$$

```
> simplify(delka-delka_upravena);
```

$$0$$

```
>
```

Bod v trojúhelníku

Uvnitř trojúhelníku s vrcholy $A = (0, 0)$, $B = (4, 0)$ a $C = (3, 2)$ najděte bod X takový, aby součet vzdáleností $AX + BX + CX$ byl minimální.

Předpokládejme, že $X = (x, y)$. Součet vzdáleností od jednotlivých vrcholů lze popsat následující funkcí:

```
> restart;
> v:=(x,y)->sqrt(x^2+y^2)+sqrt((x-4)^2+y^2)+sqrt((x-3)^2+(y-2)^2);
```

$$v := (x, y) \to \sqrt{x^2 + y^2} + \sqrt{(x-4)^2 + y^2} + \sqrt{(x-3)^2 + (y-2)^2}$$

```
> bod:=minimize(v(x,y),location);
```

$$bod := \Bigg(\Bigg(\frac{114167060}{341239663}$$

$$\mathrm{RootOf}(9409\_Z^4 - 155976\_Z^3 + 647774\_Z^2 + 156360\_Z - 2567231, 9.377772722)$$

$$-\frac{21630224}{1023718989}$$

$$\mathrm{RootOf}(9409\_Z^4 - 155976\_Z^3 + 647774\_Z^2 + 156360\_Z - 2567231, 9.377772722)^2 - \frac{790356}{341239663}$$

$$\mathrm{RootOf}(9409\_Z^4 - 155976\_Z^3 + 647774\_Z^2 + 156360\_Z - 2567231, 9.377772722)^3 + \frac{3530496874}{1023718989}\Bigg)^2 + \Bigg(-\frac{118244455}{682479326}$$

$$\mathrm{RootOf}(9409\_Z^4 - 155976\_Z^3 + 647774\_Z^2 + 156360\_Z - 2567231, 9.377772722)$$

$$+\frac{11201366}{1023718989}$$

$$\mathrm{RootOf}(9409\_Z^4 - 155976\_Z^3 + 647774\_Z^2 + 156360\_Z - 2567231, 9.377772722)^2 + \frac{818583}{682479326}$$

$$\mathrm{RootOf}(9409\_Z^4 - 155976\_Z^3 + 647774\_Z^2 + 156360\_Z - 2567231, 9.377772722)^3 + \frac{694443056}{1023718989}\Bigg)^2\Bigg)^{(1/2)} + \Bigg(\Bigg(\frac{114167060}{341239663}$$

$$\mathrm{RootOf}(9409\_Z^4 - 155976\_Z^3 + 647774\_Z^2 + 156360\_Z - 2567231, 9.377772722)$$

$$-\frac{21630224}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^2 - \frac{790356}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^3 + \frac{3530496874}{1023718989}\Bigg)^2 - \frac{913336480}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$+\frac{173041792}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^2 + \frac{6322848}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^3 - \frac{11864471168}{1023718989} + \Bigg(-\frac{118244455}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$+\frac{11201366}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^2 + \frac{818583}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^3 + \frac{694443056}{1023718989}\Bigg)^2\Bigg)^{(1/2)} + \Bigg(\Bigg(\frac{114167060}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$-\frac{21630224}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^2 - \frac{790356}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$^3 + \frac{3530496874}{1023718989}\Bigg)^2 - \frac{448513450}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$+\frac{84975880}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{2}+\frac{3104970}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{3}-\frac{10652406611}{1023718989}+\Bigg(-\frac{118244455}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)$$

$$+\frac{11201366}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{2}+\frac{818583}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{3}+\frac{694443056}{1023718989}\Bigg)^{2}\Bigg)^{(1/2)},\ \Bigg\{\Bigg[\ \Big\{x=\frac{114167060}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)$$

$$-\frac{21630224}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{2}-\frac{790356}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{3}+\frac{3530496874}{1023718989},\ y=-\frac{118244455}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)$$

$$+\frac{11201366}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{2}+\frac{818583}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)^{3}+\frac{694443056}{1023718989}\Big\},\ \Bigg(\Bigg(\frac{114167060}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4-155976\ \_Z^3+647774\ \_Z^2+156360\ \_Z-2567231,9.377772722)$$

$$-\frac{21630224}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{2} - \frac{790356}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{3} + \frac{3530496874}{1023718989}\Bigg)^2 + \Bigg(-\frac{118244455}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$+\frac{11201366}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{2} + \frac{818583}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{3} + \frac{694443056}{1023718989}\Bigg)^2\Bigg)^{(1/2)} + \Bigg(\Bigg(\frac{114167060}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$-\frac{21630224}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{2} - \frac{790356}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{3} + \frac{3530496874}{1023718989}\Bigg)^2 - \frac{913336480}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$+\frac{173041792}{1023718989}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{2} + \frac{6322848}{341239663}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$${}^{3} - \frac{11864471168}{1023718989} + \Bigg(-\frac{118244455}{682479326}$$

$$\mathrm{RootOf}(9409\ \_Z^4 - 155976\ \_Z^3 + 647774\ \_Z^2 + 156360\ \_Z - 2567231, 9.377772722)$$

$$+\frac{11201366}{1023718989}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{2}+\frac{818583}{682479326}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{3}+\frac{694443056}{1023718989}\Bigg)^{2}\Bigg)^{(1/2)}+\Bigg(\bigg(\frac{114167060}{341239663}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)$

$$-\frac{21630224}{1023718989}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{2}-\frac{790356}{341239663}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{3}+\frac{3530496874}{1023718989}\bigg)^{2}-\frac{448513450}{341239663}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)$

$$+\frac{84975880}{1023718989}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{2}+\frac{3104970}{341239663}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{3}-\frac{10652406611}{1023718989}+\bigg(-\frac{118244455}{682479326}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)$

$$+\frac{11201366}{1023718989}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{2}+\frac{818583}{682479326}$$

$\mathrm{RootOf}(9409\,\_Z^4 - 155976\,\_Z^3 + 647774\,\_Z^2 + 156360\,\_Z - 2567231, 9.377772722)^{\,}$

$$^{3}+\frac{694443056}{1023718989}\Bigg)^{2}\Bigg)^{(1/2)}\Bigg]\}$$

```
> allvalues(bod[2]);
```

$$\Bigg\{\Bigg[\Bigg\{x = \frac{4949934754}{1023718989} - \frac{21185640\sqrt{3}}{341239663} + \frac{114167060\sqrt{17+8\sqrt{3}}}{341239663}$$

$$- \frac{21630224\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$- \frac{790356\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^3}{341239663}, y = -\frac{40622989}{1023718989} + \frac{10971135\sqrt{3}}{341239663}$$

$$- \frac{118244455\sqrt{17+8\sqrt{3}}}{682479326} + \frac{11201366\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$+ \frac{818583\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^3}{682479326}\Bigg\}\Bigg], \Bigg[\Bigg(\Bigg(\frac{4949934754}{1023718989} - \frac{21185640\sqrt{3}}{341239663}$$

$$+ \frac{114167060\sqrt{17+8\sqrt{3}}}{341239663} - \frac{21630224\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$- \frac{790356\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^3}{341239663}\Bigg)^2 + \Bigg(-\frac{40622989}{1023718989} + \frac{10971135\sqrt{3}}{341239663}$$

$$- \frac{118244455\sqrt{17+8\sqrt{3}}}{682479326} + \frac{11201366\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$+ \frac{818583\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^3}{682479326}\Bigg)^2\Bigg)^{(1/2)} + \Bigg(\Bigg(\frac{4949934754}{1023718989} - \frac{21185640\sqrt{3}}{341239663}$$

$$+ \frac{114167060\sqrt{17+8\sqrt{3}}}{341239663} - \frac{21630224\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$- \frac{790356\left(\frac{402}{97} - \frac{18\sqrt{3}}{97} + \sqrt{17+8\sqrt{3}}\right)^3}{341239663}\Bigg)^2 - \frac{23219974208}{1023718989} + \frac{169485120\sqrt{3}}{341239663}$$

$$-\frac{913336480\sqrt{17+8\sqrt{3}}}{341239663}+\frac{173041792\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$+\frac{6322848\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^3}{341239663}+\Bigg(-\frac{40622989}{1023718989}+\frac{10971135\sqrt{3}}{341239663}$$

$$-\frac{118244455\sqrt{17+8\sqrt{3}}}{682479326}+\frac{11201366\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$+\frac{818583\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^3}{682479326}\Bigg)^2\Bigg)^{(1/2)}+\Bigg(\Bigg(\frac{4949934754}{1023718989}-\frac{21185640\sqrt{3}}{341239663}$$

$$+\frac{114167060\sqrt{17+8\sqrt{3}}}{341239663}-\frac{21630224\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$-\frac{790356\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^3}{341239663}\Bigg)^2-\frac{16228769711}{1023718989}+\frac{83229300\sqrt{3}}{341239663}$$

$$-\frac{448513450\sqrt{17+8\sqrt{3}}}{341239663}+\frac{84975880\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$+\frac{3104970\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^3}{341239663}+\Bigg(-\frac{40622989}{1023718989}+\frac{10971135\sqrt{3}}{341239663}$$

$$-\frac{118244455\sqrt{17+8\sqrt{3}}}{682479326}+\frac{11201366\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^2}{1023718989}$$

$$+\frac{818583\left(\frac{402}{97}-\frac{18\sqrt{3}}{97}+\sqrt{17+8\sqrt{3}}\right)^3}{682479326}\Bigg)^2\Bigg)^{(1/2)}\Bigg]\Bigg]\Bigg\}$$

```
> bod:=simplify(%);
```

$$bod := \left\{\left[\left\{x=\frac{338}{97}-\frac{112\sqrt{3}}{291},\ y=\frac{64}{97}+\frac{58\sqrt{3}}{291}\right\},\right.\right.$$

$$\frac{4\sqrt{69549-12804\sqrt{3}}}{291}+\frac{4\sqrt{6693+3492\sqrt{3}}}{291}+\frac{\sqrt{219705-76824\sqrt{3}}}{291}\Bigg]\}$$

```
> evalf(bod);
```

$$\{[\{x=2.817904843, y=1.005013563\}, 5.554854316]\}$$

```
> bod[1,1];
```

$$\{x=\frac{338}{97}-\frac{112\sqrt{3}}{291}, y=\frac{64}{97}+\frac{58\sqrt{3}}{291}\}$$

```
> assign(%);
> [x,y];
```

$$\left[\frac{338}{97}-\frac{112\sqrt{3}}{291}, \frac{64}{97}+\frac{58\sqrt{3}}{291}\right]$$

```
>
```

Obrázek:

```
> with(plots):
> with(plottools):
> p1:=polygonplot([[0,0],[4,0],[3,2]],thickness=5):
> p2:=pointplot([x,y],symbol=solidcircle,symbolsize=20,color
  =red):
> p3:=line([x,y],[0,0],color=red,thickness=3):
> p4:=line([x,y],[4,0],color=red,thickness=3):
> p5:=line([x,y],[3,2],color=red,thickness=3):
> p6:=contourplot(v(r,s),r=3/2*s..4-s/2,s=0..2,contours=[seq
  (5+i/10,i=0..25)],color=blue):
> display([p1,p2,p3,p4,p5,p6],scaling=constrained,labels=[``
  ,``]);
```




```
>
```

Poznámka:

Bod, jehož souřadnice jsme právě vypočetli, se nazývá Torricelliho bod.

>

Obraz

Na zdi je zavěšen obraz. Jeho spodní okraj A se nachází a délkových jednotek nad úrovní očí a jeho horní okraj B se nachází b délkových jednotek nad úrovní očí (viz obrázek). Určete, z jaké vzdálenosti je obraz vidět pod největším zorným úhlem.



Označme x vzdálenost od zdi (x je z intervalu $(0, nekonečno)$). Zorný úhel, pod kterým je obraz vidět, lze spočíst následovně:

```
> restart;
> assume(a>0,b>0,a<b);
> uhel:=x->arctan(b/x)-arctan(a/x);
```

$$uhel := x \rightarrow \arctan\left(\frac{b}{x}\right) - \arctan\left(\frac{a}{x}\right)$$

Stacionární bod(y):

```
> extrema(uhel(x),{},x,'stac');
```

$$\left\{-\arctan\left(\frac{-b\sim + a\sim}{2\sqrt{a\sim}\sqrt{b\sim}}\right), \arctan\left(\frac{-b\sim + a\sim}{2\sqrt{a\sim}\sqrt{b\sim}}\right)\right\}$$

```
> stac;
```

$$\{\{x = \sqrt{a\sim\, b\sim}\}, \{x = -\sqrt{a\sim\, b\sim}\}\}$$

Vidíme, že v oboru kladných čísel se nachází jeden stacionární bod $x = \sqrt{a\, b}$. Dále platí:

```
> Limit(uhel(x),x=0,right)=limit(uhel(x),x=0,right);
```

$$\lim_{x \to 0+} \arctan\left(\frac{b\sim}{x}\right) - \arctan\left(\frac{a\sim}{x}\right) = 0$$

```
> Limit(uhel(x),x=infinity)=limit(uhel(x),x=infinity);
```

$$\lim_{x \to \infty} \arctan\left(\frac{b\sim}{x}\right) - \arctan\left(\frac{a\sim}{x}\right) = 0$$

Funkční hodnota ve stacionárním bodě:

```
> hodnota:=uhel(sqrt(a*b));
```

$$hodnota := \arctan\left(\frac{b\sim}{\sqrt{a\sim\, b\sim}}\right) - \arctan\left(\frac{a\sim}{\sqrt{a\sim\, b\sim}}\right)$$

```
> simplify(hodnota);
```

$$-\arctan\left(\frac{-b\sim + a\sim}{2\sqrt{a\sim}\sqrt{b\sim}}\right)$$

```
>
```

Není těžké si rozmyslet, že hodnota v (jediném) stacionárním bodě je kladná. Vzhledem k tomu, že funkce *uhel* je diferencovatelná na intervalu (0, *nekonečno*) a limity v krajních bodech jsou nulové to znamená, že ve stacionárním bodě nastává **globální** maximum. Promyslete si to!

Abychom obraz viděli pod největším zorným úhlem, musíme se postavit do vzdálenosti $\sqrt{a\, b}$ od zdi.

```
>
```

Cvičení

1) Řešte úlohu "Krabice" v případě, že výchozí kartón má obdélníkový tvar o rozměrech 150 x 70 cm. V rozích samozřejmě stále vyřezáváme čtverce.
- Určete velikost odstřihnutých čtverců, aby vznikla krabice měla maximální objem.
- Kolik litrů se do maximální krabice vejde?

2) Na parabole $p(x) = x^2 - 6\,x + 5$ najděte bod, který má minimální vzdálenost od bodu (3, 4).
Určete i tuto (minimální) vzdálenost.
Pokud budou Vámi nalezené výsledky komplikované, zjednodušte je.

3) Jsou dány body $A = (0, 4)$ a $B = (3, 1)$.
- Vypočtěte souřadnice bodu X ležícího na ose x takového, aby součet vzdáleností $AX + BX$ byl minimální.
- Vypočtěte onen minimální součet vzdáleností.

Pokud budou Vámi nalezené výsledky komplikované, zjednodušte je.

4) Muž v loďce je vzdálen 12 km od pobřeží (majícího tvar přímky). Chce se co nejrychleji dostat do místa na pobřeží, které je od něj vzdáleno 20 km. Rozhodněte, kde se má vylodit, víte-li, že dokáže veslovat rychlostí 6 km/h a po břehu se pohybovat rychlostí 10 km/h. Určete také čas, za který se dostane do cíle (hodiny, minuty, sekundy).



>